*Replication codes for "Political Trust and Public Support for Propaganda in China" by Haifeng Huang, Chanita Intawan, and Stephen Nicholson

*Prepared in December 2023



/***** Variables

propaganda_a: support for official propaganda message A, ranging from 1 to 5
propaganda_a_scaled: propaganda_a rescaled to range from 0 to 1
propaganda_b: support for official propaganda message A, ranging from 1 to 5 
propaganda_b_scaled: propaganda_b rescaled to range from 0 to 1 
propaganda_combined: average of propaganda_a and propaganda_b
propaganda_index_rnd: propaganda_combined rounded, ranging from 1 to 5  
propaganda_scaled: propaganda_combined rescaled to range from 0 to 1

internet_a: support for internet propaganda message A, ranging from 1 to 4
internet_a_scaled: internet_a rescaled to range from 0 to 1
internet_b: support for internet propaganda message B, ranging from 1 to 4
internet_b_scaled: internet_b rescaled to range from 0 to 1
internet_c: support for internet propaganda message C, ranging from 1 to 4
internet_c_scaled: internet_c rescaled to range from 0 to 1
internet_combined: average of internet_a, internet_b, and internet_c
internet_rnd: internet_combined rounded, ranging from 1 to 4
internet_scaled: internet_combined rescaled to range from 0 to 1

diatgov: IAT D-scores, measuring implicit trust in government
diatgov_scaled: IAT D-scores rescaled to range from 0 to 1
trust_index: explicit trust, ranging from 0 to 100
trustindex_scaled: trust_index rescaled to range from 0 to 1

female: 0 or 1
agegroup: age group, ranging from 1 to 10
agegroup_scaled: agegroup rescaled to range from 0 to 1
education: education level, ranging from 1 (primary school or below) to 6 (graduate school)
education_scaled: education rescaled to range from 0 to 1
urban: urban residency (0 or 1)
income: family income status, ranging from 1 to 9
income_scaled: income rescaled to range from 0 to 1
ccpmember: membership in the Chinese Communist Party (0 or 1)
life: life satisfication, ranging from 1 to 5
life_scaled: life rescaled to range from 0 to 1
pinterest: political interest, ranging from 1 to 4
pinterest_scaled: pinterest rescaled to range from 0 to 1

*****/



clear

use trust_propaganda_repdata.dta


***** Figure 1 and Table D1 *****

grstyle init
grstyle set plain
grstyle color background white
grstyle clockdir legend_position 6
grstyle linestyle legend none

eststo clear

eststo p1:  quietly reg propaganda_a_scaled diatgov_scaled trustindex_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo p2:  quietly reg propaganda_b_scaled diatgov_scaled trustindex_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo p_all:  quietly reg propaganda_scaled diatgov_scaled trustindex_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

*Figure 1*

coefplot (p1), bylabel(Official Propaganda A) || (p2), bylabel(Official Propaganda B) || (p_all), bylabel(Aggregate) ||, ///
drop(_cons) ///
xline(0) byopts(compact graphregion(fcolor(white)) row(1)) 

*Table D1*

esttab using officialpropaganda_ols.rtf, replace b(3) se(3) ar2 label star(* 0.05 ** 0.01 *** 0.001)  compress title(Trust in Government and Attitudes toward Official Propaganda (OLS))

eststo clear

*** margins reported in the text about Figure 1***

reg propaganda_scaled diatgov_scaled trustindex_scaled i.female agegroup_scaled education_scaled i.urban income_scaled i.ccpmember life_scaled pinterest_scaled

margins, at(diatgov_scaled=(0,1)) 

reg propaganda_scaled diatgov_scaled trustindex_scaled i.female agegroup_scaled education_scaled i.urban income_scaled i.ccpmember life_scaled pinterest_scaled

margins, at(trustindex_scaled=(0,1)) 



***** Table D2 *****

eststo clear

eststo p1_o:  oprobit propaganda_a  diatgov_scaled trustindex_scaled female agegroup education ccpmember income urban life pinterest

eststo p2_o:  oprobit propaganda_b diatgov_scaled trustindex_scaled female agegroup education ccpmember income urban life pinterest

eststo p_all_o:  oprobit propaganda_index_rnd  diatgov_scaled trustindex_scaled female agegroup education ccpmember income urban life pinterest

esttab using officialpropaganda_oprobit.rtf, replace b(3) se(3) pr2 label star(* 0.05 ** 0.01 *** 0.001)  compress title(Trust in Government and Attitudes toward Official Propaganda (Ordered Probit))

eststo clear


***** Table D3 *****


eststo clear

eststo p1_s:  quietly reg propaganda_a_scaled diatgov_scaled govtrust_a_scaled  govtrust_b_scaled  govtrust_c_scaled  govtrust_d_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo p2_s:  quietly reg propaganda_b_scaled diatgov_scaled  govtrust_a_scaled  govtrust_b_scaled  govtrust_c_scaled  govtrust_d_scaled  female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo p_all_s:  quietly reg propaganda_scaled diatgov_scaled  govtrust_a_scaled  govtrust_b_scaled  govtrust_c_scaled  govtrust_d_scaled  female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled


esttab using officialpropaganda_separatetrust.rtf, replace b(3) se(3) ar2 label star(* 0.05 ** 0.01 *** 0.001)  compress title(Disaggregated Measures of Trust in Government and Attitudes toward Official Propaganda (OLS))


eststo clear


*** Figure 2 and Table D4***

eststo clear

eststo i1:  quietly reg internet_a_scaled diatgov_scaled trustindex_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo i2:  quietly reg internet_b_scaled diatgov_scaled trustindex_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo i3:  quietly reg internet_c_scaled diatgov_scaled trustindex_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo i_all:  quietly reg internet_scaled diatgov_scaled trustindex_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

* Figure 2 *

coefplot (i1), bylabel(Internet Propaganda A) || (i2), bylabel(Internet Propaganda B) || (i3), bylabel(Internet Propaganda C) || (i_all), bylabel(Aggregate) ||, ///
drop(_cons female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled) ///
xline(0)  byopts(compact graphregion(fcolor(white)) row(2))  

* Table D4 *

esttab using internetpropaganda_ols.rtf, replace b(3) se(3) ar2 label star(* 0.05 ** 0.01 *** 0.001)  compress title(Trust in Government and Attitudes toward Internet Propaganda (OLS))


eststo clear

*** margins used in the text about Figure 2***

reg internet_scaled diatgov_scaled trustindex_scaled i.female agegroup_scaled education_scaled i.urban income_scaled i.ccpmember life_scaled pinterest_scaled

margins, at(diatgov_scaled=(0,1)) 

reg internet_scaled diatgov_scaled trustindex_scaled i.female agegroup_scaled education_scaled i.urban income_scaled i.ccpmember life_scaled pinterest_scaled

margins, at(trustindex_scaled=(0,1)) 



***** Table D5 ********

eststo clear

eststo i1_o: quietly oprobit internet_a  diatgov_scaled trustindex_scaled female agegroup education ccpmember income urban life pinterest

eststo i2_o: quietly oprobit internet_b diatgov_scaled trustindex_scaled female agegroup education ccpmember income urban life pinterest

eststo i3_o: quietly oprobit internet_c diatgov_scaled trustindex_scaled female agegroup education ccpmember income urban life pinterest

eststo i_all_o: quietly oprobit internet_rnd diatgov_scaled trustindex_scaled female agegroup education ccpmember income urban life pinterest

esttab using internetpropaganda_oprobit.rtf, replace b(3) se(3) pr2 label star(+ 0.1 * 0.05 ** 0.01 *** 0.001)  compress title(Trust in Government and Attitudes toward Internet Propaganda (Ordered Probit))

eststo clear



******* using separate trust items *******


eststo clear

eststo i1_s:  quietly reg internet_a_scaled diatgov_scaled govtrust_a_scaled  govtrust_b_scaled  govtrust_c_scaled  govtrust_d_scaled female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo i2_s:  quietly reg internet_b_scaled diatgov_scaled  govtrust_a_scaled  govtrust_b_scaled  govtrust_c_scaled  govtrust_d_scaled  female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo i3_s:  quietly reg internet_c_scaled diatgov_scaled  govtrust_a_scaled  govtrust_b_scaled  govtrust_c_scaled  govtrust_d_scaled  female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled

eststo i_all_3:  quietly reg internet_scaled diatgov_scaled  govtrust_a_scaled  govtrust_b_scaled  govtrust_c_scaled  govtrust_d_scaled  female agegroup_scaled education_scaled urban income_scaled ccpmember life_scaled pinterest_scaled


esttab using internetpropaganda_separatetrust.rtf, replace b(3) se(3) ar2 label star(* 0.05 ** 0.01 *** 0.001)  compress title(Disaggregated Measures of Trust in Government and Attitudes toward Internet Propaganda (OLS))

eststo clear

*******


clear

exit
